Pensieve header: Working in the double of the 2D pencil, as determined in “Doubling.nb”.

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\2017-06"];
```

The “degree carrier” is $\hbar$, and all “coupling constants” are proportional to it.

```
$TD = 3; ℏ /: ℏ^d_· /; d > $TD := 0;
```

The Doubled Algebra $\mathcal{U}_{\hbar\alpha\beta}$

Change relative to “Doubling.nb”: We use $t = \beta a - \alpha b$ and $T = e^{\hbar t} = A^{-1} B$ instead of b and B.

Implementing $\mathcal{U}_{\hbar\alpha\beta}$

With $q = e^{\hbar\alpha\beta}$, $A = e^{-\hbar\beta a}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\hbar\alpha\beta} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\alpha y,\ [x, y]_q = \hbar^{-1}(1 - TA^2),\ [x, a] = -\alpha x\right)$.

```
Series[h^-1 (1 - e^(h t-2 h β a)), {h, 0, 5}]
```

$$(-t + 2 a \beta) - \frac{1}{2}(t - 2 a \beta)^2 h - \frac{1}{6}(t - 2 a \beta)^3 h^2 - \frac{1}{24}(t - 2 a \beta)^4 h^3 - \frac{1}{120}(t - 2 a \beta)^5 h^4 - \frac{1}{720}(t - 2 a \beta)^6 h^5 + O[h]^6$$

```
q := Sum[(ℏ α β)^k / k!, {k, 0, $TD}];
AlgebraAtom = y | a | x;
PBWRule = {y → 1, a → 2, x → 3};
B[U@a, U@y] = -α U@y; B[U@x, U@a] = -α U@x;
B[U@x, U@y] = (q - 1) UU[y, x] + UU@Sum[-(t - 2 a β)^(k+1) ℏ^k / (k + 1)!, {k, 0, $TD}];
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ < y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {t → t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] :=
  If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[c_. * l_^n_, r___] /; FreeQ[c, AlgebraAtom] :=
  Expand[c UU[Sequence @@ Table[l, {n}], r]];
UU[c_. * l : AlgebraAtom, r___] := Expand[c U[l] ** UU[r]];
UU[c_, r___] /; FreeQ[c, AlgebraAtom] := Expand[c UU[r]];
UU[] = U[];
UU[l_Plus, r___] := UU[#, r] & /@ l;
UU[l_, r___] := UU[Expand[l], r];
```

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ħ, 0, $TD}], vs] /.
     (p_ → c_) :> c UU @@ (us^p)]
 ]
```

```
B[U@x, U@y] // Simp
```

$$\left(-t-\frac{t^2\hbar}{2}-\frac{t^3\hbar^2}{6}-\frac{t^4\hbar^3}{24}\right)U[\,]+\left(2\beta+2t\beta\hbar+t^2\beta\hbar^2+\frac{1}{3}t^3\beta\hbar^3\right)U[a]+\left(-2\beta^2\hbar-2t\beta^2\hbar^2-t^2\beta^2\hbar^3\right)U[a,a]+\left(\alpha\beta\hbar+\frac{1}{2}\alpha^2\beta^2\hbar^2+\frac{1}{6}\alpha^3\beta^3\hbar^3\right)U[y,x]+\left(\frac{4\beta^3\hbar^2}{3}+\frac{4}{3}t\beta^3\hbar^3\right)U[a,a,a]-\frac{2}{3}\beta^4\hbar^3U[a,a,a,a]$$

```
$TD = 2; z1 = U[y, y, a, a, x, x]; z2 = U[y, a, x]; z3 = U[y, y, a, x];
z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp
```

$$0$$